

```
1 ' -----
2 '       From the book Inside Active Directory, ISBN 0-201-61621-1
3 '       Copyright (C) 2002 by Addison-Wesley
4 '       Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim objDSE, strDefaultDom, strDom
14 Dim objSchema, objPartitions, objRID, objDom, objInfra
15
16 Set objDSE = GetObject("LDAP://rootDSE")
17 strDefaultDom = objDSE.Get("defaultNamingContext")
18
19 strDom = InputBox("Enter the distinguished name of a domain" & _
20 vbCrLf & "(e.g. " & strDefaultDom & ")", , strDefaultDom)
21
22 If strDom = "" Then WScript.Quit(1) 'user clicked Cancel
23
24 'Get the objects that have a pointer to each operations master
25 Set objSchema = GetObject("LDAP://" & _
26 objDSE.Get("schemaNamingContext"))
27 Set objPartitions = GetObject("LDAP://CN=Partitions," & _
28 objDSE.Get("configurationNamingContext"))
29 Set objRID = GetObject("LDAP://CN=RID Manager$,CN=System," & _
30 strDom)
31 Set objDom = GetObject("LDAP://" & strDom)
32 Set objInfra = GetObject("LDAP://CN=Infrastructure," & strDom)
33
34 WScript.Echo vbCrLf & "== PER-FOREST =="
35 Call ShowFSMO("Schema", objSchema)
36 Call ShowFSMO("Domain Naming", objPartitions)
37 WScript.Echo vbCrLf & "== PER-DOMAIN =="
38 Call ShowFSMO("RID", objRID)
39 Call ShowFSMO("PDC", objDom)
40 Call ShowFSMO("Infrastructure", objInfra)
41
42 Sub ShowFSMO(strName, objPointerObj)
43 Dim objNTDS, objDC, intGap
44
45 Set objNTDS = GetObject("LDAP://" & _
46 objPointerObj.Get("fsmoRoleOwner")) 'pointer to CN=NTDS Settings
47 Set objDC = GetObject(objNTDS.Parent) 'server object
48
49 intGap = 14 - Len(strName)
50 WScript.Echo strName & " Master: " & String(intGap, " ") & _
51 objDC.dnsHostName
52 End Sub
53
```